ECE391
Computer System Engineering

Lecture 3

Dr. Zbigniew Kalbarczyk

University of Illinois at Urbana- Champaign

Spring 2021

Lecture Topics

- Continue x86 instructions
- Conditional codes
- Control flow instructions
- Assembler conventions
- Code example

Condition Codes (in EFLAGS)

- Among others (not mentioned in this class)…

  SF: sign flag: result is negative when viewed as 2's complement data type

  ZF: zero flag: result is exactly zero

  CF: carry flag: unsigned carry or borrow occurred (or other, instruction-dependent meaning, e.g., on shifts)

  OF: overflow flag: 2's complement overflow (and other instruction-dependent meanings)

  PF: parity flag: even parity in result (even # of 1 bits)

What Instructions Set Flags (condition codes)?

- Not all instructions set flags
- Some instructions set some flags!
- Use **CMP** or **TEST** to set flags:

  CMPL %EAX, %EBX # flags ⟵ (EBX – EAX)

  TESTL %EAX, %EBX # flags ⟵ (EBX *AND* EAX)

- Note that EBX does not change in either case

- What combinations of flags are needed for unsigned/signed relationships comparator?

Control Flow Instructions (1)

- Consider two three-bit values A and B; How to decide if A<B?

	#1	#2	#3	#4	#5	#6
A	010	010	010	110	110	110
B	-000	-110	-111	-000	-011	-111
C	010	100	011	110	011	111
CF	0	1	1	0	0	1
OF	0	1	0	0	1	0
SF	0	1	0	1	0	1
unsigned <	No	Yes	Yes	No	No	Yes
signed <	No	No	No	Yes	Yes	Yes

Control Flow Instructions (2)

- Note that CF suffices for unsigned <
- What about signed < ?

		CF/OF			
		00	01	11	10
SF	0	0	1	x	0
	1	1	x	0	1

- Answer: OF *XOR* SF

Branch Mnemonics

- Unsigned comparisons: “above” and “below”
- Signed comparisons: “less” and “greater”
- Both: equal/zero

unsigned	**`jne`**	**`jb`**	**`jbe`**	**`je`**	**`jae`**	**`ja`**
relationship	≠	<	≤	=	≥	>
signed	**`jne`**	**`jl`**	**`jle`**	**`je`**	**`jge`**	**`jg`**

- forms shown are those used when disassembling
  - do not expect binary to retain your version
  - e.g., “**`jnae`**” becomes “**`jb`**”

Other Control Instructions

- Other branches
  - `jo` — jump on overflow (OF)
  - `jp` — jump on parity (PF)
  - `js` — jump on sign (SF)
  - `jmp` — unconditional jump
- **Control instructions:** subroutine call and return

```
CALL  printf        # (push EIP), EIP ← printf
CALL  *%EAX         # (push EIP), EIP ← EAX
CALL  *(%EAX)       # (push EIP), EIP ← M[EAX]

RET                 # EIP ← M[ESP], ESP ← ESP + 4
```

Stack Operations

- Push and pop supported directly by x86 ISA

```
PUSHL  %EAX     # M[ESP – 4] ← EAX, ESP ← ESP – 4

POPL  %EBP      # EBP ← M[ESP], ESP ← ESP + 4

PUSHFL          # M[ESP – 4] ← EFLAGS, ESP ← ESP – 4

POPFL           # EFLAGS ← M[ESP], ESP ← ESP + 4
```

Data Size Conversion

- These instructions extend 8- or 16-bit values to 16- or 32-bit values
- General form



- Examples

```
MOVSBL  %AH, %ECX        # ECX ← sign extend to 32-bit (AH)

MOVZWL  4(%EBP), %EAX    # EAX ← zero extend to 32-bit (M[EBP + 4])
```

Assembler Conventions

label: requires a colon, and is case-sensitive
(unlike almost anything else in assembly)

comment to end of line

/* C-style comment
… (can consist of multiple lines) */

; command separator (NOT a comment as in LC-3)

```
.string “Hello, world!”, “me”     # NUL-terminated
.byte   100, 0x30, 052            # integer constants of various sizes
.word …
.long …
.quad …
.single …                         # floating-point constants
.double …
```

- If assembly file name ends in .S (case-sensitive!), file is first passed through C’s preprocessor (#define and #include)

Code Example

- **Given:** EBX pointing to an array of structures with ECX elements in the array
- **Find:** min and max age

the structure
char* name
long age



- first, define registers
  - ESI — index into array
  - EAX — current age
  - EDX — min age seen
  - EDI — max age seen
- next, use systematic decomposition…

Code Example - Loop Design Process (1)

- What is the task to be repeated?
  - update the min/max ages
  - based on the age of a single person in the array
- What are the invariants?
  - ESI = array index of person being considered
  - EDX = min age of those earlier in array
  - EDI = max age of those earlier in array
  - ECX = size of array
- What are the stopping conditions?
  - reach the end of the array
  - i.e., ESI ≥ ECX

Code Example - Loop Design Process (2)

- What must be done when a stopping condition is met?
  - nothing! (by definition of this particular problem)
- How do we prepare for the first iteration?
  - set array index to point to first person
  - set min age to something large (or to first person's, but may not exist)
  - set max age to something small
- How do we prepare for subsequent iterations?
  - update min/max age based on current person
  - increment ESI

Code Example - Loop Body



Code Example – Assembly (1)

```
        XORL %ESI,%ESI          # init vars
        MOVL TWO_MM,%EDX
        MOVL TWO_MM+4,%EDI
LOOP:
        CMPL %ECX,%ESI      # loop test; flags set based on (esi-ecx)
        JGE   DONE
        # read the age using one memory reference.
        MOVL 4(%EBX,%ESI,8),%EAX

        CMPL  %EDI,%EAX         # check the max. age
        JLE   NOT_MAX
        MOVL  %EAX,%EDI
```

Code Example – Assembly (2)

```
NOT_MAX:
            CMPL   %EDX,%EAX          # check the min age
            JGE    NOT_MIN
            MOVL   %EAX,%EDX
NOT_MIN:

            INCL %ESI                 # loop update
            JMP   LOOP

DONE:       # more code ...

TWO_MM:     .LONG 0x7FFFFFFF, 0x80000000
```